# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: William Rea <sillywilly@gmail.com>
pkgname=wv
pkgver=1.2.9
pkgrel=4
pkgdesc="MSWord library can load and parse Word 2000, 97, 95 and 6 file formats"
arch=('i686' 'x86_64')
url="https://sourceforge.net/projects/wvware"
license=('GPL')
depends=(libgsf libpng sh)
source=(https://www.abisource.com/downloads/${pkgname}/${pkgver}/${pkgname}-${pkgver}.tar.gz)
md5sums=('dbccf2e9f747e50c913b7e3d126b73f7')
sha1sums=('db4717a151742dbdb492318f104504a92075543a')

build() {
  cd $pkgname-$pkgver

  ./configure --prefix=/usr \
    --mandir=/usr/share/man --disable-static
  sed -i -e 's/ -shared / -Wl,-O1,--as-needed\0/g' libtool
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="${pkgdir}" install
}
